Newsgroup: it.comp.lang.visual-basic
Titolo: [Stamp Grafico] Come stampare una Picture...

Buon Giorno a tutti!
una domandina facile facile...
Ho un grafico "fatto in casa", dove all'interno di una picture, traccio le
linee, metto delle etichette ecc.
Il problema è: Come stampo la picture facendo uscire anche le etichette?
Teniamo presente che la picture non fa da contenitore...
Mi sarò spiegato?
Quale anima Pia mi darà un suggerimento?
Grazie a tutti per l'attenzione!
Luciano.

una domandina facile facile...
Se sai che e' facile, allora conosci gia' la risposta...
-- 
ciao, hal

 ha scritto nel messaggio
news:tqnonv4tegla750s6...@4ax.com...


 Teniamo presente che la picture non fa da contenitore...
Tenendo presente quest'ultima cosa, temo che nemmeno un Autoredraw sulla
Picture possa aiutarti.
Immagino che le varie etichette debbano essere stampate esplicitamente, una
per una.

 per una.
Era quello che temevo :( mai possibile che non ci siamo un sistema per
aggirare il problema?
ho provato appoggiandomi al buffer, ma niente, ho provato qualche API ma
niente, e una vera scocciatura!
Comunque grazie per l'aiuto...
Ciao, Luciano.

Hei ma mica ti sarai offeso? Guarda che scherzavo...
Ciao, Luciano.

 ma niente, e una vera scocciatura!
Ci sarebbe sempre il PrinForm che stampa l'intera finestra.
Cerca in questo gruppo, anche recentemente. Mi pare ci fossero soluzioni
alternative alla PrintForm che io sconsiglierei (o le cercavo e poi non ne
ho recuperate, boh?!), ma qualcosa (API) che cattura la finestra (o zona
della stessa) e poi la stampa dopo averla messa in una picture di comodo.
Vedi anche in 
www.allapi.net
 (APIGuide)

 (APIGuide)
Ecco cosa uso io (copiato da qualche parte, lo dico perchè non ci capisco
una cippa, ma va):
Il tutto "as it is", da "interpretare (ignorare eventuali ncar FSI, FSF e
FS):
'Dichiarazioni varie
' Informazioni per gestione Catture
'---------------------------------------------------------------------------
-----------------
Public Const RASTERCAPS As Long = 38
Public Const RC_PALETTE As Long = &H100
Public Const SIZEPALETTE As Long = 104
Type PALETTEENTRY
peRed As Byte
peGreen As Byte
peBlue As Byte
peFlags As Byte
End Type
Type LOGPALETTE
palVersion As Integer
palNumEntries As Integer
palPalEntry(255) As PALETTEENTRY ' Enough for 256 colors.
End Type
Type GUID
Data1 As Long
Data2 As Integer
Data3 As Integer
Data4(7) As Byte
End Type
Type RECT
Left As Long
Top As Long
Right As Long
Bottom As Long
End Type
Type PicBmp
Size As Long
Type As Long
hBmp As Long
hPal As Long
Reserved As Long
End Type
'Sub/Functions
' Informazioni per gestione Catture
'---------------------------------------------------------------------------
-----------------
Public Function CreateBitmapPicture(ByVal hBmp As Long, ByVal hPal As Long)
As Picture
ncar = "CreateBitmapPicture"
' CreateBitmapPicture
' - Creates a bitmap type Picture object from a bitmap and
' palette.
'
' hBmp
' - Handle to a bitmap.
'
' hPal
' - Handle to a Palette.
' - Can be null if the bitmap doesn't use a palette.
'
' Returns
' - Returns a Picture object containing the bitmap.
Dim r As Long
Dim pic As PicBmp
' IPicture requires a reference to "Standard OLE Types."
Dim IPic As IPicture
Dim IID_IDispatch As GUID
' Fill in with IDispatch Interface ID.
With IID_IDispatch
.Data1 = &H20400
.Data4(0) = &HC0
.Data4(7) = &H46
End With
' Fill Pic with necessary parts.
With pic
.Size = Len(pic) ' Length of structure.
.Type = vbPicTypeBitmap ' Type of Picture (bitmap).
.hBmp = hBmp ' Handle to bitmap.
.hPal = hPal ' Handle to palette (may be null).
End With
' Create Picture object.
r = OleCreatePictureIndirect(pic, IID_IDispatch, 1, IPic)
' Return the new Picture object.
Set CreateBitmapPicture = IPic
End Function
Public Function CaptureWindow(ByVal hWndSrc As Long, ByVal Client As
Boolean, ByVal LeftSrc As Long, ByVal TopSrc As Long, ByVal WidthSrc As
Long, ByVal HeightSrc As Long) As Picture
ncar = "CaptureWindow"
' CaptureWindow
' - Captures any portion of a window.
'
' hWndSrc
' - Handle to the window to be captured.
'
' Client
' - If True CaptureWindow captures from the client area of the
' window.
' - If False CaptureWindow captures from the entire window.
'
' LeftSrc, TopSrc, WidthSrc, HeightSrc
' - Specify the portion of the window to capture.
' - Dimensions need to be specified in pixels.
'
' Returns
' - Returns a Picture object containing a bitmap of the specified
' portion of the window that was captured.
Dim hDCMemory As Long
Dim hBmp As Long
Dim hBmpPrev As Long
Dim r As Long
Dim hDCSrc As Long
Dim hPal As Long
Dim hPalPrev As Long
Dim RasterCapsScrn As Long
Dim HasPaletteScrn As Long
Dim PaletteSizeScrn As Long
Dim LogPal As LOGPALETTE
' Depending on the value of Client get the proper device context.
If Client Then
hDCSrc = GetDC(hWndSrc) ' Get device context for client area.
Else
hDCSrc = GetWindowDC(hWndSrc) ' Get device context for entire
' window.
End If
' Create a memory device context for the copy process.
hDCMemory = CreateCompatibleDC(hDCSrc)
' Create a bitmap and place it in the memory DC.
hBmp = CreateCompatibleBitmap(hDCSrc, WidthSrc, HeightSrc)
hBmpPrev = SelectObject(hDCMemory, hBmp)
' Get screen properties.
RasterCapsScrn = GetDeviceCaps(hDCSrc, RASTERCAPS) ' Raster
' capabilities.
HasPaletteScrn = RasterCapsScrn And RC_PALETTE ' Palette
' support.
PaletteSizeScrn = GetDeviceCaps(hDCSrc, SIZEPALETTE) ' Size of
' palette.
' If the screen has a palette make a copy and realize it.
If HasPaletteScrn And (PaletteSizeScrn = 256) Then
' Create a copy of the system palette.
LogPal.palVersion = &H300
LogPal.palNumEntries = 256
r = GetSystemPaletteEntries(hDCSrc, 0, 256, LogPal.palPalEntry(0))
hPal = CreatePalette(LogPal)
' Select the new palette into the memory DC and realize it.
hPalPrev = SelectPalette(hDCMemory, hPal, 0)
r = RealizePalette(hDCMemory)
End If
' Copy the on-screen image into the memory DC.
r = BitBlt(hDCMemory, 0, 0, WidthSrc, HeightSrc, hDCSrc, LeftSrc, TopSrc,
vbSrcCopy)
' Remove the new copy of the on-screen image.
hBmp = SelectObject(hDCMemory, hBmpPrev)
' If the screen has a palette get back the palette that was
' selected in previously.
If HasPaletteScrn And (PaletteSizeScrn = 256) Then
hPal = SelectPalette(hDCMemory, hPalPrev, 0)
End If
' Release the device context resources back to the system.
r = DeleteDC(hDCMemory)
r = ReleaseDC(hWndSrc, hDCSrc)
' Call CreateBitmapPicture to create a picture object from the
' bitmap and palette handles. Then return the resulting picture
' object.
Set CaptureWindow = CreateBitmapPicture(hBmp, hPal)
End Function
Public Function CaptureScreen() As Picture
ncar = "CaptureScreen"
' CaptureScreen
' - Captures the entire screen.
'
' Returns
' - Returns a Picture object containing a bitmap of the screen.
Dim hWndScreen As Long
' Get a handle to the desktop window.
hWndScreen = GetDesktopWindow()
' Call CaptureWindow to capture the entire desktop give the handle
' and return the resulting Picture object.
Set CaptureScreen = CaptureWindow(hWndScreen, False, 0, 0, Screen.Width \
Screen.TwipsPerPixelX, Screen.Height \ Screen.TwipsPerPixelY)
End Function
Public Function CaptureForm(frmSrc As Form) As Picture
ncar = "CaptureForm"
' CaptureForm
' - Captures an entire form including title bar and border.
'
' frmSrc
' - The Form object to capture.
'
' Returns
' - Returns a Picture object containing a bitmap of the entire
' form.
' Call CaptureWindow to capture the entire form given its window
' handle and then return the resulting Picture object.
Set CaptureForm = CaptureWindow(frmSrc.hWnd, False, 0, 0,
frmSrc.ScaleX(frmSrc.Width, vbTwips, vbPixels), frmSrc.ScaleY(frmSrc.Height,
vbTwips, vbPixels))
End Function
Public Function CaptureClient(frmSrc As Form) As Picture
ncar = "CaptureClient"
' CaptureClient
' - Captures the client area of a form.
'
' frmSrc
' - The Form object to capture.
'
' Returns
' - Returns a Picture object containing a bitmap of the form's
' client area.
' Call CaptureWindow to capture the client area of the form given
' its window handle and return the resulting Picture object.
Set CaptureClient = CaptureWindow(frmSrc.hWnd, True, 0, 0,
frmSrc.ScaleX(frmSrc.ScaleWidth, frmSrc.ScaleMode, vbPixels),
frmSrc.ScaleY(frmSrc.ScaleHeight, frmSrc.ScaleMode, vbPixels))
End Function
Public Function CaptureActiveWindow() As Picture
ncar = "CaptureActiveWindow"
' CaptureActiveWindow
' - Captures the currently active window on the screen.
'
' Returns
' - Returns a Picture object containing a bitmap of the active
' window.
Dim hWndActive As Long
Dim r As Long
Dim RectActive As RECT
' Get a handle to the active/foreground window.
hWndActive = GetForegroundWindow()
' Get the dimensions of the window.
r = GetWindowRect(hWndActive, RectActive)
' Call CaptureWindow to capture the active window given its
' handle and return the Resulting Picture object.
Set CaptureActiveWindow = CaptureWindow(hWndActive, False, 0, 0,
RectActive.Right - RectActive.Left, RectActive.Bottom - RectActive.Top)
End Function
Public Sub PrintPictureToFitPage(Prn As Object, pic As Picture, Optional
parametri = 0)
ncar = "PrintPictureToFitPage"
' PrintPictureToFitPage
' - Prints a Picture object as big as possible.
'
' Prn
' - Destination Printer object.
'
' Pic
' - Source Picture object.
' parametri 2^0 Stampa senza ampliare
Const vbHiMetric As Integer = 8
Dim PicRatio As Double
Dim PrnWidth As Double
Dim PrnHeight As Double
Dim PrnRatio As Double
Dim PrnPicWidth As Double
Dim PrnPicHeight As Double
If (parametri And 2 ^ 0) Then
Prn.PaintPicture pic, 0, 0
GoTo esce
End If
' Determine if picture should be printed in landscape or portrait
' and set the orientation.
'If Prn Is Printer Then
' If Pic.Height >= Pic.Width Then
' Prn.Orientation = vbPRORPortrait ' Taller than wide.
' Else
' Prn.Orientation = vbPRORLandscape ' Wider than tall.
' End If
'End If
' Calculate device independent Width-to-Height ratio for picture.
PicRatio = pic.Width / pic.Height
' Calculate the dimentions of the printable area in HiMetric.
PrnWidth = Prn.ScaleX(Prn.ScaleWidth, Prn.ScaleMode, vbHiMetric)
PrnHeight = Prn.ScaleY(Prn.ScaleHeight, Prn.ScaleMode, vbHiMetric)
' Calculate device independent Width to Height ratio for printer.
PrnRatio = PrnWidth / PrnHeight
' Scale the output to the printable area.
If PicRatio >= PrnRatio Then
' Scale picture to fit full width of printable area.
PrnPicWidth = Prn.ScaleX(PrnWidth, vbHiMetric, Prn.ScaleMode)
PrnPicHeight = Prn.ScaleY(PrnWidth / PicRatio, vbHiMetric, Prn.ScaleMode)
Else
' Scale picture to fit full height of printable area.
PrnPicHeight = Prn.ScaleY(PrnHeight, vbHiMetric, Prn.ScaleMode)
PrnPicWidth = Prn.ScaleX(PrnHeight * PicRatio, vbHiMetric, Prn.ScaleMode)
End If
' Print the picture using the PaintPicture method.
Prn.PaintPicture pic, 0, 0, PrnPicWidth, PrnPicHeight
esce:
End Sub
'Nel chiamante (Chiaramente da "interpretare" e adattare)
'-----------------------------------
' Per catturare la finestra
Screen.ActiveForm.Refresh
Set ESF0MESS.pic_catture = CaptureForm(Screen.ActiveForm)
'ESF0MESS.pic_catture è una PictureBox
stop
' Per stampare quanto catturato
amplia = F_MESSAGGIO_MODALE("Ampliare / Adeguare in funzione" & ACAPO &
"delle dimensioni al foglio?", vbYesNoCancel + vbQuestion)
If amplia <> vbCancel Then
Call F_MESSAGGIO("Stampa in corso...", 2 ^ 13)
Call PrintPictureToFitPage(INFO_STAMPA.dest, ESF0MESS.pic_catture,
IIf(amplia = vbYes, 0, 2 ^ 0)) 'Printer
Call P_FINE_STAMPA
Call F_TOGLI_MESSAGGIO(2 ^ 13)
End If
stop

 capisco una cippa, ma va):
Forse da qui:
http://support.microsoft.com/default.aspx?scid=kb;en-us;161299

 Ci sarebbe sempre il PrinForm che stampa l'intera finestra.
E noi usiamo il cervellino :)
Se lo fa la printform prima di tutto significa che si può fare.
Ora la domanda: si può fare in VB?
Credo che sia tutto sommato banale (poi magari non è vero :P )
una BitBlt() che "taglia" l'hDC del form e lo appiccica sull'hDC della
printer.
Ciao
--
Math Parser : 
http://www.neodatatype.net
Sito Comune : 
http://www.it-lang-vb.net